
Heterogeneous Volume Modeling 

Aim:  The study of the process of the heterogeneous volume modeling.      

Task:  Create multimaterial 3D models using the geometric modeling language and system 

HyperFun.  

Result: The source code. The report.   

 

Theoretical part: 
 

For modeling and fabricate any nature objects the new model is required. It should allow to set and 

combine geometry of object and it’s properties. Physical and abstract objects and phenomena are 

modeled as point sets. Properties are modeled as attributes. Examples of attributes: 

 photometric (opacity, color, reflectance) 

 material 

 density 

 temperature 

 etc. 

 

Requirements to a model 

 Separation of geometry and attributes 

 Photometric, physical, other attributes 

 Uniform treatment of geometry and attributes 

 Constructive modeling of both point set geometry and attributes using primitives, operations, 

and relations  



 Discrete and continuous fields 

 Space-time and multidimensional modeling 

 

Model of heterogeneous object: point set and point attributes  

Hypervolume is a multidimensional point set with multiple attributes 

 

O = ( F(X), S1 (X), …, Sk (X) ) 

F(X) - FRep of geometry 

Si (X) – attributes based on FRep  

            space partition 

 

Application heterogeneous volume model: 

 Internal structure with non-uniform distribution of material and other attributes 

 entities of different dimensionalities   

 varying material distribution in CAD/CAM and rapid prototyping 

 physical simulations, geological and medical modeling, volume modeling and rendering 

 Fabrication of objects with multiple materials and varying material distribution; 

 

 

 

 

 

 

 

 

 

 

 

 

 



Example №1 of heterogeneous volume in Hyperfun: 

Block with different materials 

 

“my_model” 

--This object is provided as a test. 

-- The program contains the definitions of two objects 

-- Definition of object Color Sphere 

ColorSphere(x[3], a[7], s[3]) { 

array center[3], rgb[3]; 

center[1] = a[1]; center[2] = a[2]; center[3] = a[3]; 

radius = a[4]; 

rgb[1] = a[5]; rgb[2] = a[6]; rgb[3] = a[7]; 

-- s   = [0, 0, 0]; 

sphere = hfSphere(x, center, radius); 

  

-- association the sphereв_Ts interior and surface with the color 

        if (sphere >= 0) then 

   s[1] = rgb[1]; s[2] = rgb[2]; s[3] = rgb[3]; 

        endif; 

  

ColorSphere = sphere; 

} 

-- Definition of object "my_model" using previously defined ColorSphere 

my_model(x[3], a[1], s[3]) { 

array BlockCenter[3]; 



array rgb1[3], rgb2[3]; 

array SpherePar[7]; 

array s1[4]; 

rgb1= [1,0,0]; 

rgb2= [0,0,1]; 

 

 s = [0.0,1.0,0.0];  

SpherePar = [0,0,0,3,0,0,0]; 

SphereRadius = a[1]; 

BlockCenter = [-7,-7,-7]; 

model = hfBlock(x, Blockcenter, 14, 7, 14); 

  

if(model > -0.1) then 

         SpherePar[5] = rgb1[1]; SpherePar[6] = rgb1[2]; SpherePar[7] = rgb1[3]; 

 tmp = ColorSphere(x, SpherePar, s); 

   

 SpherePar[1]= 7; 

  SpherePar[5] = rgb2[1]; SpherePar[6] = rgb2[2]; SpherePar[7] = rgb2[3]; 

  tmp = ColorSphere(x, SpherePar, s); 

  

endif;   

my_model = model;  

}  

  



 

Example №2 of heterogeneous volume in Hyperfun: 

Model with attributes 

       

“my_model”   “my_partition” 

 

--This object is provided as a test. 

--This HyperFun program consists of one geometric object: 

--union of superellipsoid, torus and soft object  

--Below two space paritions for attributes s[] 

--First paritition 

-- consisting in a union of blocks (5x5x5) 

-- between blocks, attribute is a blue color 

-- inside the block, attribute is a blue color 

--Second parition 

-- consisting in a torus 

my_partition(x[3],a[1],s[3]) 

{ 

array xxt[3],center[3]; 

---------------- 

--first parition 

 blocks = -1; --function value for the block partition 

i=0; 

j=0; 



k=0; 

dx = 3.35;  -- size of the bricks in x direction 

dy = 3.35;  -- size of the bricks in y direction 

dz = 3.35;  -- size of the bricks in z direction 

NN = 5;     -- size of the grid of blocks: NNxNNxNN 

  while (i<NN) loop     --loop along x axis 

    j=0; 

    while (j<NN) loop     --loop along y axis      

      k=0; 

      while (k<NN) loop     --loop along z axis 

        --definition of the block center 

        center[1] = -10 + i*(4); 

        center[2] = -6 + j*(4);          

        center[3] = -10 + k*(4);   

        --union of the blocks 

        blocks = blocks | hfBlock(x,center,dx,dy,dz); 

        k =k+1; 

      endloop;   

      j=j+1; 

    endloop; 

    i=i+1; 

  endloop; 

  --default color: mortar 

  s[1] = 0.0; 

  s[2] = 0.0; 

  s[3] = 1.0; 

  -- color of the briks 

  if(blocks>=0.0) then 

    s[1] = 1.0; 

    s[2] = 0.0; 

    s[3] = 0.0; 

  endif; 



 

---------------- 

--second parition 

-- torus is a partition corresponding to the torus in the geometry model 

-- torus is defined by library function: a bit larger than the one defined in my_object 

  center = [0, -9, 0];  

  torus = hfTorusY(x,center,3.5,1.2); 

 

 if(torus > -0.01) then 

    s[1] = hfA_Wave(x[1],0,1,1); 

    s[2] = hfA_Wave(x[2],0.2,0.8,2); 

    s[3] = 0.2; 

  endif; 

 my_partition = blocks | torus; 

} 

my_model(x[3], a[1], s[3]) 

{ 

array x0[9], y0[9], z0[9], d[9], center[3]; 

x1=x[1]; 

x2=x[2]; 

x3=x[3]; 

-- superellipsoid by formula 

superEll = 1-(x1/0.8)^4-(x2/10)^4-(x3/0.8)^4; 

-- torus by library function 

center = [0, -9, 0];  

torus = hfTorusY(x,center,3.5,1); 

 

-- soft object  

x0 = [2.,1.4, -1.4, -3, -3, 0, 2.5, 5., 6.5]; 

y0 = [8, 8, 8, 6.5, 5, 4.5, 3, 2, 1]; 

z0 = [0, -1.4,-1.4, 0, 3, 4, 2.5, 0, -1]; 

d = [2.5, 2.5, 2.5, 2.5, 2.5, 2.5, 2.5, 2.7, 3]; 



sum = 0.; 

i = 1; 

while (i<10) loop  

     xt = x[1] - x0[i];  

     yt = x[2] - y0[i];  

     zt = x[3] - z0[i];  

     r = sqrt(xt*xt+yt*yt+zt*zt); 

     if (r <= d[i]) then  

          r2 = r*r; r4 = r2*r2; r6 = r4*r2;  

          d2 = d[i]^2; d4 = d2*d2; d6 = d4*d2;  

          sum = sum + (1 - 22*r2/(9*d2) + 17*r4/(9*d4) - 4*r6/(9*d6));  

     endif;  

     i = i+1;  

endloop;  

soft = sum - 0.2;  

model =  superEll | torus | soft; 

--define the space partition around this model and attributes obtained in s[] array 

if ( model > -0.1) then 

  tmpP = my_partition(x,a,s); 

endif; 

-- final geometry model with attributes returned in s[] array 

my_model =model ;  

} 

 

Tasks: 
 

 Create your own heterogeneous volume object using Hyperfun. 

 


